

APPENDIX VI

ARM'S-LENGTH SALES TO AFFILIATED PARTIES

In order to report sales to an affiliated comparison market customer (“affiliate”) rather than resales to unaffiliated customers by that affiliated comparison market customer, a respondent must demonstrate that the sales to the affiliated comparison market customer are at arm’s length. This appendix describes the basis for this analysis, and provides guidance as to how a respondent could go about demonstrating that sales to an affiliated comparison market customer are at arm’s length.

Overview

In accordance with section 351.403(c) of the Department’s regulations, we include home market or third-country affiliated party sales in our analysis only if the respondent’s sales are made at “arm’s length.” To be at “arms-length” the prices of the affiliated-party transactions must be comparable to the prices at which the respondent sold identical merchandise to unaffiliated parties. In determining whether affiliated party transactions are made at arm's-length prices, we generally compare the respondent’s reported prices to affiliated parties with the respondent’s prices to unaffiliated parties at the same level of trade. In making such a comparison, the Department has established a ratio in which sales by the exporter or producer to an affiliate be included in the normal value calculation. For affiliated party sales to be considered in the normal value calculation, prices to an affiliate must be between 98 percent and 102 percent, inclusive, of prices to unaffiliated customers. If affiliated party prices are, on average, less than 98 percent or more than 102 percent of unaffiliated party prices, then we reject them. In establishing the 98/102 percent band, the Department aims to prevent distortion of normal value based on sales between affiliates which could be unreasonably high or low-priced. Instead, the ratio is set up to capture only those sales between affiliates which are made in the ordinary course of trade. See Antidumping Proceedings: Affiliated Party Sales in the Ordinary Course of Trade, 67 FR 69187 (November 15, 2002).

Transactions Covered

For investigations, the databases used for the arm’s length test are limited to all sale observations with sale dates that fall within the POI. For administrative reviews, the databases used for the arm’s-length test are limited to sales observations with sale dates that fall within the appropriate reporting period for the comparison market.

For both investigations and administrative reviews, the databases are also limited to those transactions involving merchandise described in Appendix III of the questionnaire.

Net Price Calculations

The Department’s arm’s-length test adjusts gross unit prices for relevant adjustments (*e.g.*, for appropriate discounts and rebates, movement expenses, imputed credit

expenses, other direct selling expenses, commissions, packing, *etc.*), and, for each affiliated party comparison market customer, compares the net prices, by CONNUM, of sales to the affiliated party to the net prices of sales to all unaffiliated parties.

In instances where the respondent is required to respond to Section D of the questionnaire and where certain CONNUMs were sold to an affiliated comparison market customer but not to unaffiliated comparison market customers, the comparisons are made to the most similar product sold to unaffiliated comparison market customers, with appropriate adjustments made for differences in merchandise (the appropriate variable and total cost data for each comparison market CONNUM are therefore required for this calculation to be made). The determination of what constitutes the most similar merchandise is based upon the hierarchy of the product characteristics, and of the subcategories within each characteristic, identified in Section B of the questionnaire.

Comparisons are not allowed across levels of trade, or of prime merchandise to non-prime merchandise, or across manufacturers (MFRH), or, in instances where the respondent is not required to respond to Section D of the questionnaire, for CONNUMs sold to an affiliated comparison market customer but not to unaffiliated comparison market customers. In such instances where comparison are not allowed, the sales to the affiliated comparison market customers are ignored for purposes of the results of the arm's length test.

As noted above, for affiliated party sales to be considered in the normal value calculation, prices to an affiliate must on average be between 98 percent and 102 percent, inclusive, of prices to unaffiliated customers.

Arm's-Length Test Programming

The arm's-length test programming language currently used by the Department is referenced below. To demonstrate that sales to an affiliate were at arm's length, you should utilize the methodology reflected in this programming, in conjunction with your appropriate databases and following the guidance above with respect to appropriate sales observations, adjustments, *etc.* (CMSALES would be the file of all comparison market POI sales of merchandise described in Appendix III of the questionnaire, and &CMAFFL &CMLOT &CMMANF &CMPRIME &CMCUST are references to CUSRELH, LOTH, MFRH, PRIMEH (for those cases in which PRIMEH is included in Section B of the Department's questionnaire), and CUSCODH, respectively):

```
%MACRO CM4_ARMSLEN;
  %IF &RUN_ARMSLEN = YES %THEN
  %DO;
    **-----**
    ** CM4-A: Check Customer Affiliation and LOT **
    **-----**

    PROC SORT DATA = CMSALES;
      BY &CMAFFL &CMLOT &CMMANF &CMPRIME &CMCUST;
    RUN;
```

VI-3

```

PROC MEANS NOPRINT DATA = CMSALES;
  BY &CMAFFL &CMLOT &CMMANF &CMPRIME ;
  VAR &CMQTY;
  OUTPUT OUT = TOTS (DROP=_TYPE_ _FREQ_)
    N=SALES SUM=TOTQTY;
RUN;

PROC SORT DATA = CMSALES OUT = NUMCUST NODUPKEY;
  BY &CMAFFL &CMLOT &CMMANF &CMPRIME &CMCUST;
RUN;

PROC SUMMARY NOPRINT DATA = NUMCUST;
  BY &CMAFFL &CMLOT &CMMANF &CMPRIME;
  OUTPUT OUT = NUMCUST (DROP=_TYPE_);
RUN;

DATA TOTCUST;
  MERGE TOTS NUMCUST (RENAME=( _FREQ_ =NCUST));
  BY &CMAFFL &CMLOT &CMMANF &CMPRIME ;
RUN;

PROC PRINT DATA = TOTCUST SPLIT='*';
  VAR &CMAFFL &CMLOT &CMMANF &CMPRIME  NCUST SALES TOTQTY;
  SUM NCUST SALES TOTQTY;
  FORMAT SALES COMMA7. TOTQTY COMMA12.;
  LABEL &CMAFFL = "CUSTOMER *AFFILIATION*======"
    &CMLOT = "LEVEL(S)*OF TRADE*======"
    &CMMANF = "MANUFACTURER*CODE *======"
    &CMPRIME = "PRIME &*SECONDS*======"
    NCUST = "NUMBER OF*CUSTOMERS*======"
    SALES = "TOTAL*SALES*======"
    TOTQTY = "TOTAL *QUANTITY*======" ;
  TITLE3 'SUMMARY OF CUSTOMER AFFILIATION & LOT';
RUN;

**-----**;
** CM4-B: Weighted-average net prices **;
** of affiliated customers **;
**-----**;

PROC SORT DATA = CMAFF;
  BY &CMCUST &CMLOT &CMMANF &CMPRIME &CMCONNUM;
RUN;

PROC MEANS NOPRINT DATA = CMAFF;
  BY &CMCUST &CMLOT &CMMANF &CMPRIME &CMCONNUM;
  ID &CMVCOM &CMTCOM &CMCHAR;
  VAR CMNETPRI;
  WEIGHT &CMQTY;
  OUTPUT OUT=TOTAFF (DROP=_FREQ_ _TYPE_)
    N=AFFOBS SUMWGT=AFFQTY MEAN=AFFNETPR;
RUN;

PROC PRINT DATA=TOTAFF (OBS=&PRINT25);
  TITLE3 'SAMPLE OF AFFILIATED WEIGHTED-AVERAGE NET PRICES';
RUN;

```

VI-4

```

PROC SORT DATA=TOTAFF;
  BY &CMLLOT &CMMANF &CMPRIME &CMCONNUM &CMCUST;
RUN;
**-----**;
** CM4-C: Weighted-average net prices **;
**       of non-affiliated customers **;
**-----**;

PROC SORT DATA = CMNAF;
  BY &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
RUN;

PROC MEANS NOPRINT DATA = CMNAF;
  BY &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
  ID NAFLOT NAFMANF NAFPRIME NAFCONN NAFVCOMH &NAFCHAR;
  VAR CMNETPRI;
  WEIGHT &CMQTY;
  OUTPUT OUT=TOTNAF (DROP=_FREQ_ _TYPE_)
         N=NAFOBS SUMWGT=NAFQTY MEAN=NAFNETPR;
RUN;

PROC PRINT DATA=TOTNAF (OBS=&PRINT25);
  TITLE3 'SAMPLE OF UNAFFILIATED WEIGHTED-AVERAGE NET PRICES';
RUN;

**-----**;
** CM4-D: Identify identical product matches at the same LOT **;
**-----**;

DATA ALLCOMP NOCOMP;
  MERGE TOTAFF (IN=A) TOTNAF (IN=B);
  BY &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
  IF A AND B THEN
  DO;
    NAFADJPR = NAFNETPR;
    MATCH    = 'IDENTICAL';
    DIFMER   = 0;
    COSTDIFF = 0;
    OUTPUT ALLCOMP;
  END;
  ELSE IF A & NOT B THEN OUTPUT NOCOMP;
RUN;

**-----**;
** CM4-E: Identify similar product matches at the **;
**       same LOT only if cost data are available **;
**-----**;

%MACRO CM4_SIM;
  %IF &CHK4SIM = YES %THEN
  %DO;
    DATA SIMMATCH;
      SET NOCOMP;

      DO J=1 TO LAST;
        SET TOTNAF (DROP=&CMLLOT &CMMANF &CMPRIME

```

VI-5

```

                                &CMCONNUM)
POINT=J NOBS=LAST;

LOTDIFF = ABS(NAFLOT - &CMLOT);
IF &CMVCOM GT . AND NAFVCOMH GT .
THEN DIFMER = NAFVCOMH - &CMVCOM;
ELSE DIFMER = .;

IF DIFMER GT .
THEN COSTDIFF = ABS(DIFMER/&CMTCOM);
ELSE COSTDIFF = .;

**-----**
** CM4-F Examine all product matches within the **
**      20% DIFMER test and at the same LOT      **
**-----**

IF (0.20 GE COSTDIFF GT .) &
  LOTDIFF = 0 &
  &CMMANF = NAFMANF &
  &CMPRIME = NAFPRIME THEN
DO;
  ARRAY AFFCHR (*) &CMCHAR;
  ARRAY NAFCHR (*) &NAFCHAR;
  ARRAY DIFCHR (&CMNUMCHR) ;

  **-----**
  ** CM4-G: Select similar products based **
  **      on product characteristics.    **
  **-----**

  DO I=1 TO DIM(DIFCHR);
    DIFCHR(I) = ABS(AFFCHR(I)-NAFCHR(I));
  END;
  DROP I;
  MATCH = 'SIMILAR';
  NAFADJPR = NAFNETPR - DIFMER;
  OUTPUT SIMMATCH;
END;
END;
RUN;

PROC SORT DATA=SIMMATCH;
  BY &CMLOT &CMMANF &CMPRIME &CMCUST &CMCONNUM
  MATCH DIFCHR1-DIFCHR&CMNUMCHR LOTDIFF COSTDIFF;
RUN;

DATA SIMCOMP (DROP=DIFCHR1-DIFCHR&CMNUMCHR) TOP5SIM
  SIM5 (KEEP=&CMLOT &CMMANF &CMPRIME &CMCUST &CMCONNUM);
SET SIMMATCH;
BY &CMLOT &CMMANF &CMPRIME &CMCUST &CMCONNUM
  MATCH DIFCHR1-DIFCHR&CMNUMCHR LOTDIFF COSTDIFF;

DIFCODE = CATX("-", OF DIFCHR1-DIFCHR&CMNUMCHR);
IF FIRST.&CMCONNUM THEN CHOICE = 0;
CHOICE + 1;
IF CHOICE = 1 THEN OUTPUT SIMCOMP;

```

VI-6

```

        IF CHOICE LE 5 THEN OUTPUT TOP5SIM;
        IF CHOICE = 5 THEN OUTPUT SIM5;
RUN;

PROC DATASETS LIBRARY=WORK NOLIST;
    DELETE SIMMATCH;
QUIT;

DATA SIMPRT;
    MERGE TOP5SIM (IN=A) SIM5 (IN=B);
    BY &CMLLOT &CMMANF &CMPRIME &CMCUST &CMCONNUM;
    IF A & B THEN OUTPUT SIMPRT;
RUN;

PROC PRINT DATA = SIMPRT (OBS = 15);
    BY &CMLLOT &CMMANF &CMPRIME &CMCUST &CMCONNUM;
    PAGEBY &CMCONNUM;
    VAR &CMCHAR NAFLOT NAFCONN &NAFCHAR
        &CMVCOM &CMTCOM NAFVCOMH DIFMER
        MATCH DIFCODE LOTDIFF COSTDIFF CHOICE;
    TITLE3 'CHECK TOP 5 SIMILAR MATCHES (NAFADJPR INCLUDES DIFMER)';
RUN;

**-----**;
** CM4-H: Combine identical and similar matches **;
**-----**;

DATA ALLCOMP;
    SET ALLCOMP SIMCOMP;
    DROP LOTDIFF DIFCODE CHOICE;
RUN;
%END;
%MEND CM4_SIM;

%CM4_SIM;

**-----**;
** CM4-I: For each product sold to an affiliate with an **;
** identical or similar match, calculate the percent **;
** ratio (PCTRATIO) of the weighted-average affiliated **;
** price to the weighted-average non-affiliated price. **;
**-----**;

PROC SORT DATA = TOTAFF;
    BY &CMCUST &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
RUN;

PROC SORT DATA = ALLCOMP;
    BY &CMCUST &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
RUN;

DATA ALLCOMP;
    MERGE TOTAFF (IN=A) ALLCOMP (IN=B);
    BY &CMCUST &CMLLOT &CMMANF &CMPRIME &CMCONNUM;
    IF A;
    IF MATCH = '' THEN MATCH = 'NONE';
    IF NAFADJPR = .

```

VI-7

```

THEN PCTRATIO = .;
ELSE PCTRATIO=(AFFNETPR/NAFADJPR)*100;
RUN;

PROC SORT DATA = ALLCOMP;
  BY &CMCUST &CMLOT &CMMANF &CMPRIME &CMCONNUM;
RUN;

PROC PRINT DATA=ALLCOMP (OBS=&PRINT50); ** (OBS=&PRINT50);
  BY &CMCUST;
  ID &CMCUST;
  VAR &CMLOT &CMMANF &CMPRIME &CMCONNUM AFFOBS AFFQTY AFFNETPR &CMVCOM &CMTCOM
      NAFLOT NAFCONN NAFOBS NAFQTY NAFNETPR NAFVCOMH
      MATCH DIFMER COSTDIFF NAFADJPR PCTRATIO;
  TITLE3 'SAMPLE RATIOS OF AFFILIATED TO UNAFFILIATED PRICES';
RUN;

**-----**
** CM4-J: Calculate the overall CUSRATIO for **
**     each affiliated customer, and keep **
**     affiliated customers with CUSRATIOS **
**     at or between 98 - 102 percent.   **
**-----**

PROC MEANS NOPRINT DATA = ALLCOMP;
  BY &CMCUST;
  VAR PCTRATIO;
  WEIGHT AFFQTY;
  OUTPUT OUT=AFFCUST (DROP=_FREQ_ _TYPE_)
          N=NUMCOMP MEAN=CUSRATIO;
RUN;

PROC MEANS NOPRINT DATA = CMAFF;
  BY &CMCUST;
  OUTPUT OUT = AFFOBS (DROP=_FREQ_ _TYPE_) N=AFFOBS;
RUN;

DATA AFFCUST;
  MERGE AFFCUST (IN=A) AFFOBS (IN=B);
  BY &CMCUST;
  IF A & B;
  IF 98 LE CUSRATIO LE 102 THEN
  DO;
    RESULTS = 'PASS';
    PASSOBS = AFFOBS;
  END;
  ELSE
  DO;
    RESULTS = 'FAIL';
    FAILOBS = AFFOBS;
  END;
RUN;

PROC SORT DATA=AFFCUST;
  BY &CMCUST;
RUN;

```

VI-8

```
PROC PRINT DATA=AFFCUST (OBS=45) SPLIT='*';
  VAR &CMCUST AFFOBS NUMCOMP CUSRATIO RESULTS;
  SUM NUMCOMP AFFOBS PASSOBS FAILOBS;
  LABEL &CMCUST = 'AFFILIATED*CUSTOMER(S)*===== '
        AFFOBS = 'CUSTOMER *SALES *===== '
        NUMCOMP = 'CUSTOMER *WT-AVG PRICES*COMPARED *===== '
        CUSRATIO = 'CUSTOMER *WT-AVG PRICE*RATIO *===== '
        RESULTS = 'CUSTOMER*TEST *RESULTS*===== '
        PASSOBS = 'TOTAL SALES *PASSING TEST*===== '
        FAILOBS = 'TOTAL SALES *FAILING TEST*===== ';
  FORMAT AFFOBS NUMCOMP PASSOBS FAILOBS COMMA12.;
  TITLE3 "TEST RESULTS FOR AFFILIATED CUSTOMER(S)";
  RUN;
```